

PLOT



A PLOTTING SYSTEM FOR THE EXIDY SORCERER by Vic Tolomei

A
product
of



QUALITY SOFTWARE

Low Resolution Plotting	page 2
Loading LORES	page 3
LORES Demo	page 4
High Resolution Plotting	page 5
Loading HIRES	page 7
HIRES Demo	page 7

LOW RESOLUTION PLOTTING

LORES is a low resolution plotting package for the Sorcerer which is similar to LORES graphics on the Apple II, but in black and white. It consists of several machine language programs that can be called easily from BASIC. It allows a program to plot, anywhere on the screen,

1. vertical lines,
2. horizontal lines, and
3. points

using any of the hundreds of characters available on the Sorcerer. There is a version of LORES for each size Sorcerer available. The differences will be explained later. First a description of how to do BASIC low resolution graphics.

The LORES routines require the use of special BASIC variables for communications. These are

XS — determines the horizontal size of the screen window used for the plot

YS — determines the vertical size of the screen window

CL — determines the character to be used for a line/point

AA, BB, CC, XX, YY — depends on the plotting function being performed (see steps 5, 6, and 7).


Here are simple step-by-step instructions:

1. The first four statements of your BASIC program should be typed in according to the directions for your size Sorcerer (see page 4). For example, for a 16K Sorcerer:

```
0 GOTO 4
1 POKE260,135:POKE261,59:Z9=USR(Z9):RETURN
2 POKE260,170:POKE261,59:Z9=USR(Z9):RETURN
3 POKE260,213:POKE261,59:Z9=USR(Z9):RETURN
```

Statement 1 is a GOSUB subroutine which calls the PLOT (plot a point) LORES program. Statement 2 calls VLIN (draw a vertical line) and 3 calls HLIN (draw a horizontal line).

2. Decide on the size of the screen "window" in which you want to plot. The largest size is the size of the actual TV screen, but smaller sizes can be chosen and LORES will center the window in the TV screen. The size is chosen by setting the variables XS and YS. For example, to have a 40 point wide by 10 point high window —



enter the statement:

```
4 XS=39:YS=9
```

To use the entire TV screen (64 by 30):


```
4 XS=63:YS=29
```

So XS can be any integer from 0 through 63, and YS 0 through 29. LORES will not accept any other values.

Note that XS and YS are actually the numbers of the last points on a line or column, which is one less than the real size.

LOW RESOLUTION PLOTTING, CONTINUED

- Decide on a character to be used for the plotting. This can be done once for the entire plot, or can be changed at any time for each point or line. For example:
5 CL=ASC("A")
will plot A's while
5 CL=192
(192 is the first programmable character) will plot the character you program at memory locations hex FE00 to FE07.
- Clear the screen:
6 PRINT CHR\$(12)
- TO PLOT A POINT ON THE SCREEN
—Set XX to the horizontal position of the point on the screen *window* (from 0 to XS) and set YY to the vertical position (0 to YS). Note: upper left corner is (0,0)
—Set the plot character CL (if necessary)
—GOSUB1
For example, on the 40 by 10 window described in step 2, to plot a point in the third space to the right and fifth down:
7 XX=2:YY=4:GOSUB1



- TO DRAW A VERTICAL LINE ON THE SCREEN
—Decide the horizontal position for the line; set CC to that value (0 thru XS)
—Set AA to the first vertical position and BB to the last vertical position of the line (both are 0 thru YS, and BB must be greater than or equal to AA)
—Set character CL (if necessary)
GOSUB2
For example, to draw two vertical lines down the sides of the 40x10 window above
8 CC=0:AA=0:BB=9:GOSUB2
9 CC=39:AA=0:BB=9:GOSUB2
- TO DRAW A HORIZONTAL LINE ON THE SCREEN
—Decide the vertical position for the line; set CC to that value (0 thru YS)
—Set AA to the first horizontal position and BB to the last horizontal position of the line (both are 0 thru XS, and BB must be greater than or equal to AA)
—Set character CL (if necessary)
—GOSUB3
For example, to draw two horizontal lines across the top and the bottom of the 40x10 window above
10 CC=0:AA=0:BB=39:GOSUB3
11 CC=9:AA=0:BB=39:GOSUB3

LOADING LORES

Once your program is written load the LORES package appropriate for your Sorcerer size (see below) and run your program.

For example, for a 16K Sorcerer (information from the computer is in italics):

```
READY
BYE
>LOAD LOW16
    FOUND - etc
    LOADING - etc
>PP
READY
RUN
```

To run any BASIC program thereafter which uses LORES, the LORES subroutine package must always be loaded into memory before the BASIC program is actually "RUN."

LORES FOR AN 8K SORCERER

Statements 0 through 3 of your BASIC program should be:

```
0 GOTO 4
1 POKE260,135:POKE261,27:Z9=USR(Z9):RETURN
2 POKE260,170:POKE261,27:Z9=USR(Z9):RETURN
3 POKE260,213:POKE261,27:Z9=USR(Z9):RETURN
```

The name of the LORES file on the tape is LOW8.

LORES FOR A 16K SORCERER

Everything here is pretty much the same as it was for the 8K version of LORES above, except for the following points:

1. In statements 1 through 3, the "27" in each of the second POKE's must be changed to a "59"
2. The file name for the 16K version of LORES on the cassette tape is LOW16

Of course the 8K version of LORES will work on a 16K Sorcerer, but it does not use memory as efficiently as LOW16.

LORES DEMO

On the cassette tape is included a demonstration BASIC program which uses the LORES package. This program is written for an 8K Sorcerer so it will run on any machine. The name of the file is DEMOL, and it should be run as all LORES programs, as follows:

```
(Reset the EXIDY)
READY
BYE
>LO LOW8

    FOUND — etc
    LOADING — etc
>PP
READY
CLOAD DEMOL
    FOUND — etc
    LOADING — etc
READY
RUN
```

We hope you enjoy using LORES. It should make the tedious job of doing low resolution graphics just a little bit easier and therefore a little bit more fun and productive.

HIGH RESOLUTION PLOTTING

The Exidy Sorcerer has a feature fairly unique among current microcomputer systems: the user-programmable character set. The programmer can choose any of 64 or 128 characters, each of which is 8 dots wide and 8 high. Any of these 64 small dots can be turned on or off to form any shape or character desired. Since the video screen is 64 by 30 characters in size, that provides well over 100,000 points.

Using this feature, high resolution graphics can be done on the Sorcerer. To plot some points, programmable characters are defined with each point represented by an "on" dot in one of these characters. These characters are then placed on the screen at the appropriate locations. Thus graphs, curves, and lines of very high accuracy and resolution can be drawn.

HIRES is a package of several machine language subroutines written for the Sorcerer and callable from BASIC. The BASIC program simply calculates the (x,y) coordinate pairs for each point (dot) and HIRES will take care of the rest.

The HIRES screen is 512 dots across and 240 dots down. The lower left hand corner is designated (0,0), as shown here.



To plot points on this screen grid, the BASIC programmer must set some variables with the information needed, and call the HIRES subroutines (GOSUB). Follow these simple step-by-step instructions:

1. The first three BASIC statements of your program must be entered according to your Sorcerer memory size. The actual statements are described later. For example, for a 16K Sorcerer:

```
0 GOTO100
1 POKE260,0:POKE261,60:Z9=USR(Z9):RETURN
10 POKE260,18:POKE261,60:Z9=USR(Z9):RETURN
```

Statement 1 calls the HIRES subroutine NEWP to plot a new graph on the screen. Statement 10 calls the HIRES subroutine ADD to put additional points on an existing plot.

2. Decide how many points are to be plotted, and set the variable PN to that value. For example, to plot 300 points:

```
100 PN=300
```

HIGH RESOLUTION PLOTTING, CONTINUED

3. Define two arrays with PN elements in each. Remember that Sorcerer BASIC uses subscript 0. One array is called PX and will contain all the "x" (horizontal) values of points, while the other is PY and will contain all the "y" (vertical) values. E.g.:

```
110 DIM PX(PN-1),PY(PN-1)
```

Note again that the subscripts start at 0 and run through PN-1.

4. Fill the PX and PY arrays with the PN (x,y) pairs for the PN points to be plotted. It is important that each PX value be an integer no less than 0 and no greater than 511, and each PY value also be an integer between 0 and 239. HIRES will not accept any other values, and ignores all fractional parts of non-integral (floating point) numbers. To draw a sine wave:

```
120 A=30: REM AMPLITUDE IS 30 DOTS HIGH
130 F=120: REM VERTICAL DISPLACEMENT IS UP 120 DOTS
140 FOR I=0 TO PN-1
150 PX(I)=I
160 PY(I)=A*SIN(PX(I)*.05)+F
170 NEXT I
```

5. Call the HIRES routine NEWP to put this graph on the screen. NEWP will first clear the screen, erase the cursor, and then draw the plot described by PX and PY almost instantaneously:

```
180 GOSUB 1
```

6. To add more points to this graph after it is already on the screen (for example for plot superimposure), the ADD HIRES routine should be used. Set a new PN if necessary, and put any new or additional points in the PX and PY arrays, and finally call the ADD subroutine:

```
190 FOR I=0 TO PN-1
200 PX(I)=I+20
210 NEXT I
220 GOSUB 10
```

This takes the sine wave already on the screen, and plots another identical one on top of it, but shifted to the right 20 dots.

- Note: HIRES has only 128 characters to "program" or define for plotting. The other 128 are reserved by the Sorcerer for control characters and standard ASCII alphanumerics and punctuation. If HIRES exhausts these characters, it will put an "O" (Overflow) in the lower left corner of the screen to indicate this condition, and it will not perform any more superimposure (ADD) plots. Either further plots will have to use the NEWP function (and thus use new program-mable characters), or the plot which caused the overflow will have to be reduced in size (thus using fewer characters).

LOADING HIRES

Once the program is completely written, load the HIRES package appropriate for your size Sorcerer, and run it.

For example, for a 16K Sorcerer (information from the computer is in italics):

```
READY
BYE
>LOAD HI16
    FOUND — etc
    LOADING — etc
>PP
READY
RUN
```

To run any subsequent BASIC program which uses HIRES, make sure the HIRES package is loaded into memory before the "RUN" command is entered.

HIRES FOR 8K AND LARGER SORCERERS

Statements 0, 1 and 10 of your BASIC program should be:

```
0 GOTO100
1 POKE260,0:POKE261,28:Z9=USR(Z9):RETURN
10 POKE260,18:POKE261,28:Z9=USR(Z9):RETURN
```

The name of the HIRES file on the cassette tape is HI8.

HIRES FOR 16K AND LARGER SORCERERS

As with the LORES package, 16K HIRES is pretty much identical to 8K HIRES, except for the following:

1. The "28" in statements 1 and 10 above should be changed to a "60."
2. The cassette tape filename for 16K HIRES is HI16.

8K HIRES will work just fine on a 16K or larger EXIDY, but it will not use memory as efficiently for some applications.

HIRES DEMO

On the cassette tape is included a BASIC program written to demonstrate and use the HIRES subroutines. This program is written for 8K machines, so it should run on any Sorcerer. The file name is "DEMOH," and running it is the same as running any HIRES program:

```
(reset the machine)
READY
BYE
>LO HI8
    FOUND — etc
    LOADING — etc
>PP
READY
CLOAD DEMOH
    FOUND — etc
    LOADING — etc
READY
RUN
```

PLOT

PLOT is a self-contained and complete plotting package which allows the Sorcerer BASIC programmer to easily perform complex high resolution or low resolution graphics on the Sorcerer.

Included on cassette are two completely separate and independent machine language subroutines that allow high or low resolution plotting, plus two pre-written BASIC demonstration programs to illustrate the use of the plotting subroutines. In this instruction manual you will find complete documentation, including examples, on how the plotting routines can be called easily from a BASIC program.

Enjoy our other SORCERER software

FASTGAMMON by Bob Christiansen. Our popular machine language backgammon game that started us in business. The computer plays against you and makes good moves instantaneously. Option to replay dice rolls from the previous game. An eight-page instruction booklet is included.

On cassette, **\$19.95**

~~On diskette, \$24.95~~

Z-80 DISASSEMBLER by Vic Tolomei. Decode machine language programs, including SORCERER'S monitor and ROM-PACS, with this Z-80 Disassembler written in BASIC. Instruction mode prints out standard Z-80 mnemonics. Or use ASCII mode which converts machine code to ASCII. **\$14.95**

MAGIC MAZE™ by Vic Tolomei. A challenging maze game. Ten levels of play. Holding your lantern, you wander through a maze trying to stay on the right path and avoid pitfalls. Automatic scoring tells you how good a pathfinder you are. **\$11.95**

SOFTWARE INTERNALS MANUAL FOR THE SORCERER by Vic Tolomei. A must for anyone writing software for the SORCERER. Seven chapters: Intro to Machine Language, Devices & Ports, The Monitor, Cassette Interface, BASIC structure, Video & Graphics, The Keyboard. Indexed. Includes diagrams and software routines. 64 pages. **\$14.95**

FOR LATEST RELEASES, WRITE FOR OUR CATALOG



QUALITY SOFTWARE

6660 Reseda Blvd., Suite 103, Reseda, Ca. 91335

(213) 344-6599